

DEPARTMENT OF MECHANICAL ENGINEERING & MECHANICS
COLLEGE OF ENGINEERING & TECHNOLOGY
OLD DOMINION UNIVERSITY
NORFOLK, VIRGINIA 23529

1N-09-CR

48376

P. 28

**FURTHER DEVELOPMENTS RELATING TO THE
NASA LANGLEY RESEARCH CENTER 13-INCH
MAGNETIC SUSPENSION AND BALANCE SYSTEM**

By

Colin P. Britcher, Principal Investigator

Final Report

For the period ended July 31, 1991

Prepared for

National Aeronautics and Space Administration

Langley Research Center

Hampton, Virginia 23665

Under

Research Grant NAG-1-1142

David A. Dress, Technical Monitor

AAD-Super/Hyper Aerodynamics Branch

(NASA-CR-188995) FURTHER DEVELOPMENTS
RELATING TO THE NASA LANGLEY RESEARCH CENTER
13-INCH MAGNETIC SUSPENSION AND BALANCE
SYSTEM Final Report, period ending 31 Jul.
1991 (Old Dominion Univ.) 28 p CSCL 14B G3/09

N92-11023

Unclass
0048376

November 1991

Old Dominion University Research Foundation is a not-for-profit corporation closely affiliated with Old Dominion University and serves as the University's fiscal and administrative agent for sponsored programs.

Any questions or comments concerning the material contained in this report should be addressed to:

Executive Director
Old Dominion University Research Foundation
P. O. Box 6369
Norfolk, Virginia 23508-0369

Telephone: (804) 683-4293
FAX Number: (804) 683-5290

DEPARTMENT OF MECHANICAL ENGINEERING & MECHANICS
COLLEGE OF ENGINEERING & TECHNOLOGY
OLD DOMINION UNIVERSITY
NORFOLK, VIRGINIA 23529

**FURTHER DEVELOPMENTS RELATING TO THE
NASA LANGLEY RESEARCH CENTER 13-INCH
MAGNETIC SUSPENSION AND BALANCE SYSTEM**


By

Colin P. Britcher, Principal Investigator

Final Report
For the period ended July 31, 1991

Prepared for
National Aeronautics and Space Administration
Langley Research Center
Hampton, Virginia 23665

Under
Research Grant NAG-1-1142
David A. Dress, Technical Monitor
AAD-Super/Hyper Aerodynamics Branch



November 1991

Contents

1. Introduction
2. Modifications to the Digital Controller
 - 2.1. Introduction
 - 2.2. Revised Vertical Position Sensing System
 - 2.2.1. Physical Arrangement
 - 2.2.2. Interfacing and Data Acquisition
 - 2.2.3. Edge Processing
 - 2.3. Revised CPU
 - 2.4. Integrator Jitter
3. Base Pressure Telemetry from a Magnetically Suspended Wind Tunnel Model
 - 3.1. Introduction
 - 3.2. Telemetry System Design
 - 3.3. Transducer and Telemetry Calibration
 - 3.4. Test Procedure
4. Revisions to the Wind Tunnel Test Section

References

Appendix A Revised Software Listings

DIGSCN

CNTROL - Version 1

CNTROL - Version 2

CNTROL - Version 3

1. Introduction

Since 1979, when the 13-inch MSBS was moved to Langley Research Center, a steady development effort, aimed at improving all aspects of the system hardware has been proceeding. This report concentrates on a few specific developments that have been undertaken recently :-

- a) Modifications to the digital control system to accommodate a modified position sensing system
- b) Development of pressure telemetry systems
- c) Revisions to the wind tunnel test section

2. Modifications to the digital controller

2.1. Introduction

During 1990, a modified position sensing system was developed for the 13-inch MSBS by the Instrument Research Division at NASA LaRC [1,2]. The principal improvement from the existing system was the use of larger and more dense photodiode arrays (4096 instead of 1024 elements and $15\mu\text{m}$ instead of $25.4\mu\text{m}$ spacing) in the "vertical" channels, i.e. those sensing vertical translation and pitch rotation. This was primarily intended to permit suspension of models over a larger range of angles of attack. Since the width (in the vertical direction) of the light beam is now roughly 6 cms. (instead of roughly 2.5 cms.), edge detection circuitry was developed to sense two light/dark transitions, one normally corresponding to the top of the model, the other to the bottom. With both edges now detectable, it happens that more slender models can now be suspended, i.e. models of substantially less than the previous minimum of 2.5 cms. diameter.

Fairly extensive changes to the control system software and some changes in input interfacing were necessary to accommodate and fully exploit this new sensing capability. Further, it had also become clear that the interfacing of the controller to the MSBS hardware through I/O devices is not optimal. This is partly due to the original desire to accommodate system growth, such as extra electromagnets, bipolar power supplies and so on. Problems that have arisen include a slight "jitter" of the suspended model when the error integrators are active and awkward scaling of gains and position attitude demands. In addition, since the controller was last fully documented, the CPU

has been changed from a PDP 11/23+ to a PDP 11/73. This section of this report will detail the changes made to the controller software and hardware for all the reasons stated above. The baseline configuration is as detailed in Reference 3.

2.2. Revised Vertical Position Sensing System

2.2.1 Physical Arrangement

Reticon 4096 element photodiode arrays (RL4096) are mounted vertically alongside the test section. The functional beam layout resembles the previous layout though with considerable changes in the way the beams are split, expanded and collimated, as shown in Figure 1. The naming and numbering conventions for the sensors have been changed, due to revision of interface electronics.

Channel	Register	Name	Sensor location	Function
1	2	AX	Front or back	Axial motion
2	4	FU	Above nose	Lateral motion of front of model
3	3	AU	Above tail	Lateral motion of aft of model
4	5	FL	Alongside nose	Vertical motion of front of model
5	1	AL	Alongside tail	Vertical motion of aft of model

Table 1 - Naming and numbering conventions for position sensor channels

Note that the CPU register usage has not been changed, so that the old channel numbers correspond to the current register numbers.

The new arrays, Channels 4 and 5, detect and store (internally) 2 edge locations. These locations are multiplexed onto common data lines with the data selection controlled from the host computer via a single dataline. Due to the inability of the DRV-11 digital I/O interface to assert individual I/O bits without corrupting the state of other tri-state lines, the signal bit is derived from a spare ADV-11 D/A converter channel, as illustrated in Figure 2. With this device, the low 4 bits of the 12 to be converted are made available as digital I/O bits. Bit 0 is used, so that the future use of the D/A is compromised to the least extent possible.

The physical arrangement of the sensors results in scanning starting from the top of the array and proceeding downwards. In normal operation (with both edges of the model visible) the video signal would contain a light → dark → light sequence. If one or

other edge is not visible (located off the end of the array), the following truth table applies:

Edges Visible	Signal Bit	Output
Both	0	Light → dark location (top of model)
	1	Dark → light location (bottom of model)
Only light → dark (model low)	0	Light → dark location
	1	0 pixel count
Only dark → light (model high)	0	0 pixel count
	1	Dark → light location
Neither	0	0 pixel count
	1	0 pixel count

It is clear that some care is required when operating close the bottom end of the array, when the Dark → light transition not only becomes undetected, but reverts to a zero count, representing the top of the array. There is no corresponding difficulty at the top of the array.

2.2.2 Interfacing and Data Acquisition

With two 4096 element arrays and three 1024 element arrays, there are 54 bits of sensor data, instead of 50 as previously. The I/O port allocations have been completely changed as a result :

Sensor	Name	I/O port
1	AX	A, bits 0-9
2	FU	A, bits 10-15 plus B, bits 0-3
3	AU	B, bits 4-13
4	FL	C, bits 0-11
5	AL	C, bits 12-15 plus D, bits 0-7

Subroutine DIGSCN, part of the CNTRLB library, has been completely revised to reflect these changes and also to read in all available data (i.e. both sets of edge data from the new vertical arrays). Since extra edge data is available, the allocation of locations in the CHARS array has changed. The raw data from the new arrays is stored in a sequence of locations previously reserved for electromagnet current information,

ASCII labels 6 through 9. Once the appropriate processing has been completed (in the main body of the CNTROL subroutine), final data is transferred to the old locations, ASCII labels SP (AL sensor) and \$ (FL sensor). A listing of the revised subroutine is included in Appendix A.

2.2.3 Edge Processing

There are at least three possibilities. Following the traditional mode of operation, the bottom edge only could be monitored. This form of CNTROL is rather straightforward, only changed in detail from the old version. This is included as version 1 of CNTROL in Appendix A.

Equally straightforwardly, the top edge only could be sensed. This is version 2 of CNTROL, also included in Appendix A. Neither of these two options fully exploits the new capabilities of the new sensors and both have some limits on the allowable range of motion of the model.

The more general solution is to monitor both top and bottom edges of the model. When both are visible, it is logical to combine the two in order to generate an estimate of the location of the model's centroid. This would automatically accommodate rolling non-axisymmetric models or axially translating tapered models, as illustrated in Figure 3, provided that both edges remain in the field of view. Of course, more sophisticated processing will still be required for more complex geometries.

If one edge moves out of range, there is no reason to abandon control. Rather, the remaining edge should be monitored, with the location of the model's centroid estimated from previous data concerning the width (the difference between the two visible edge locations), if required for control purposes. When the second edge moves back into view, normal operation resumes.

Based on previous experience of "IF---THEN---ELSE" logic decisions in real-time control loops, it was decided that the change of edge choice and processing should not be made abruptly. Rather, as one edge approaches its respective end of the array, the system should "phase out" that edge. With both edges in the field of view, the governing equations are as follows :

$$\text{Centroid location} = \frac{1}{2} (\text{Edge}_1 + \text{Edge}_2) \quad \text{Model width} = (\text{Edge}_1 - \text{Edge}_2)$$

- where Edge_n refers to the relevant pixel count and Edge_1 is the light \rightarrow dark transition

(the top of the model). If the model becomes very "low" in the beam, Edge₂ would approach the end of the array, so a weighting factor is introduced :

$$\text{Centroid} = W_1 \times (\text{Edge}_1 + \frac{\text{Width}}{2}) + W_2 \times (\text{Edge}_2 - \frac{\text{Width}}{2})$$

Model width = Held at previous value

Similar procedures are employed if the model becomes very "high" in the beam. The general idea is illustrated in Figure 4. The exact interval of phase-out used is rather arbitrary. A third version of CNTROL has been written and is also listed in Appendix A, but has not yet been fully debugged or tested, due to lack of time.

2.3. Revised CPU

The old PDP 11/23 + CPU was operating at or above 90% capacity as the MSBS controller, with a 256Hz clock rate. Clearly, there would have been serious problems if any extra operations were required in the controller. The PDP 11/73 upgrade was a plug-compatible replacement. The measured performance increase in this application (biggest increment in floating point operations), is just less than a factor of 2, so that the controller is now operating at about 50% CPU capacity.

2.4. Integrator Jitter

It has been observed that when the suspended model settles to its steady-state location, excessive "jitter" appears to arise from the action of the error integrators. The motion is at a low amplitude and frequency, less than 1 Hz. Monitoring of compensator, integrator and controller outputs indicates that the software is functioning as intended. However, as the integrator accumulator increments and decrements by 1 LSB, the step function to the power supplies is just large enough to cause observable motion. The controller output contains higher amplitude steps, of course, but the frequencies involved are much higher.

The problem can be alleviated by re-scaling variables in the control system, in the following way. The D/A output ranges for the 4 "vertical" electromagnets should be changed, from $\pm 10V$ at present, to $+10V$ (Bipolar operation of these electromagnets is not likely in the near future). This will reduce the current step corresponding to 1 LSB by a factor of 2. Gain values used in all degrees of freedom, except axial, would then be raised by a factor of 2. Additionally, the OUTPUT subroutine used for these four

electromagnets would have to be re-coded, so as to correct the zero-current settings and offsets therefrom.

3. Base Pressure Telemetry from a Magnetically Suspended Wind Tunnel Model

3.1. Introduction

Telemetry of base pressure information from a magnetically suspended wind tunnel model was first demonstrated as practical by researchers at ONERA, Chatillon, France, where base pressures from an axisymmetric model in supersonic flow were transmitted by R.F. link [4,5,6]. Later, similar measurements were attempted at RAE, Farnborough, England. The range of pressures encountered in supersonic flow is quite large, though it appears that sufficient accuracy was still difficult to achieve.

A miniature single-channel pressure telemetry system for use in magnetically suspended models in subsonic flows was developed by Tcheng, Schott and Bryant [7]. This paper will discuss the use of this system in the NASA Langley 13 inch MSBS and will review modifications made and practical difficulties encountered.

3.2. Telemetry System Design

MSBS models have to contain a magnetic core of some type. A cylindrical soft-iron core is used in the 13-inch MSBS. To provide the maximum magnetic force and moment capability for a given size model, the core occupies a large fraction of the available volume and represents the majority of the model's weight. Due to these constraints on available volume and weight, a simple circuit design was chosen, based on four low-power operational amplifiers and a voltage to frequency convertor. The variable frequency output signal triggers short-duration flashes of a miniature infra-red Light-Emitting-Diode. The general circuitry and component layout are detailed in Figure 5. More details can be found in References 7,8.

Following previous practice, the initial design utilized an absolute pressure transducer so that base pressure could be measured directly, with no resort to a reference pressure. Due to the small pressure variations encountered in low-speed flows, a low range absolute transducer was created by filling the back cavity of a sensitive differential transducer with inert gas (argon) and sealing the vent tube. Thus the gas reservoir served as the transducer reference pressure.

In practice, this arrangement was most unsatisfactory. As is common with most wind tunnel testing, the stream temperature is poorly regulated. In the case of the 13-inch MSBS, the tunnel has an outside atmospheric return. With the tunnel stopped, test section temperature drifts towards laboratory ambient. With the flow started, the temperature rapidly reaches outside ambient. The inevitable temperature variations thus created resulted in large apparent zero shifts in the transducer. Since the base pressure coefficients being measured were close to zero, the measurement uncertainty was very large.

$$C_{P_b} = \frac{p_b - p_s}{\frac{1}{2}\rho U^2} = \frac{(p_b - p_{r_2}) - (p_{r_1} - p_{atm}) - (p_{atm} - p_s)}{\frac{1}{2}\rho U^2}$$

- where p_r is the transducer reference pressure, $(p_{r_1} - p_{atm})$ is measured with the tunnel stopped, $(p_{atm} - p_s)$ is deduced from tunnel calibration (and can be assumed accurate here) and $(p_b - p_{r_2})$ is the run-time measurement. If $p_{r_1} = p_{r_2}$ then there is no particular problem. However, the reference pressure is temperature sensitive according to :

$$p_{r_2} = p_{r_1} \left(\frac{T_2}{T_1} \right), \quad \text{so that } \Delta C_{P_b} = \frac{p_{r_1} (T_2 - T_1)}{\frac{1}{2}\rho U^2 (T_1)}$$

Assuming standard atmospheric initial conditions, the error as a function of tunnel speed is shown in Figure 6. It is clearly unacceptably large, particularly at the lower speeds.

For later tests, the absolute transducer was replaced by a sensitive differential transducer. Conventional wind tunnel procedure would now dictate that the reference side of the transducer be connected to test section static pressure. This is impossible here, since a stable source of this pressure signal is inaccessible. On the other hand, test section Total (Pitot) pressure is relatively easily obtained, by using a pitot tap positioned somewhere on the model.

$$C_{P_b} = \frac{p_b - p_s}{\frac{1}{2}\rho U^2} = \frac{(p_b - p_o) - (p_o - p_s)}{\frac{1}{2}\rho U^2}$$

- where $(p_b - p_o)$ is the run-time measurement and $(p_o - p_s)$ is derived from tunnel calibration.

A transducer "zero" is still taken, with the tunnel stopped, so temperature sensitivity is still important, although not explicitly shown in the above equation.

3.3. Transducer and Telemetry Calibration

Due to the problems mentioned above, extensive calibration of the differential transducer assembly was undertaken. This included pressure calibration with temperature, supply voltage and magnetic field variations. Results are summarized in Figures 7-9. Since battery pack capacity was between 150 and 200 mA, depending on choice of cell, it is clear that with the typical current drain of around 40mA, adequate working time can be achieved with no significant sensitivity shifts. There is a small residual temperature sensitivity, apparently originating principally in the electronics package. Curve fits from Figure 9 result in the following calibration equations:

$$\begin{aligned}\text{Freq (in kHz)} &= 10.4483 - 1.0505 \times 10^{-3} \text{ Pascals ; } T_{\text{ambient}} = 25^{\circ}\text{C} \\ \text{Freq (in kHz)} &= 10.6995 - 1.0794 \times 10^{-3} \text{ Pascals ; } T_{\text{ambient}} = 38.9^{\circ}\text{C}\end{aligned}$$

It should be noted that the largest effect is a zero shift and can therefore be corrected for to some extent in tunnel testing.

The C_{p_b} error due to temperature uncertainties and residual sensitivity is shown in Figure 10, and is seen to be greatly improved compared to the previous version.

3.4. Test Procedure

One base pressure tap is selected and the remainder sealed with adhesive tape. The telemetry system is activated by connecting the battery pack and securing the nosepiece. Transmission is continuous until the battery is disconnected or exhausted. The model is placed in the wind tunnel on the mechanical launching device and a low-speed flow started with the objective of thermally conditioning the model and telemetry system. The MSBS is activated and the model "launched" and maneuvered to the desired position and attitude in the test section. The tunnel flow is now stopped and a "zero" frequency recorded (with zero pressure differential). The tunnel flow is now brought rapidly to the first desired speed and the transmitted frequency recorded again, with simple averaging over a few seconds to overcome any unsteadiness. Further tunnel speeds are selected as desired. The tunnel is now stopped and the zero frequency recorded again. Any significant discrepancy between the first and second "zeros" is assumed to be due to insufficient thermal conditioning or falling battery voltage and the whole test sequence is disregarded. The model can now be removed from the tunnel for selection of a different pressure tap and the whole process repeated. With experience, around ten pressure taps can be tested at four different tunnel speeds within two hours.

4. Revisions to the Wind Tunnel Test Section

Various problems with the original test section, such as a slight longitudinal pressure gradient, optical degradation of the "Lexan" material (prompting the installation of glass windows) and structural flexibility, led to a decision to completely redesign the test section and associated diffuser. Full details of the new design will not be presented here, but the important changes will be summarized.

The basic material will be fiberglass of approximately 0.5 inch thickness. Metal window frames will be set into the lay-up during manufacture. Each frame is gapped at one corner to eliminate eddy currents. Optical glass windows will be mounted on metal sub-frames, similarly gapped. The test section inner dimensions match the original at the upstream end [10], but are slightly enlarged at the downstream end to accommodate boundary layer growth. The joint between the test section and the diffuser will be changed to a bolted-flange type to simplify removal of the tunnel components from the MSBS. A woven wire catch net will be incorporated into the diffuser to prevent model flyaways from reaching the turning vanes ahead of the fan. The model launching devices will be mated to the test section in the same way as at present, but will require redesign and refabrication since the new optical system intrudes into the previous areas occupied by the launching arm mechanisms.

References

1. Schott, T.D.; Tchong, P.: A high-resolution electro-optical displacement measuring system. ISA Symposium, Orlando, FL, 1989. ISA 89-0045.
2. Schott, T.D.; Tchong, P.: The new electro-optical displacement measuring system for the 13-inch MSBS. Workshop on Aerospace Applications of Magnetic Suspension Technology. NASA Langley Research Center, September 1990. NASA CP-10066, March 1991.
3. Britcher, C.P.: User guide for the digital control system of the NASA Langley Research Center's 13-inch Magnetic Suspension and Balance System. NASA CR-178210, March 1987.
4. Beasssier, J.: Telemetry foe a Model Magnetically Suspended in a Wind Tunnel. La Recherche Aeronautique, May-June 1961. In French.

5. Dubois, G.: Drag of Wind Tunnel Models of Various Ogival Forms Suspended Magnetically. La Recherche Aeronautique, March-April 1962. (see NASA TM 77325 for translation)
6. Moreau, R.: Use of Magnetic Suspension System in O.N.E.R.A. Wind Tunnel. ARL Symposium on Magnetic Wind Tunnel Model Suspension and Balance Systems. April 1966.
7. Tcheng, P.; Schott, T.D.; Bryant, L.: A Miniature Infrared Pressure Telemetry System. 34th International Instrumentation Symposium, Albuquerque, NM, May 1988.
8. Alcorn, C.W.: An Experimental Investigation of the Aerodynamic Characteristics of Slanted-Base Ogive-Cylinders using Magnetic Suspension Technology. NASA CR 181708, November 1988.
9. Britcher, C.P.; Alcorn, C.W.; Kilgore, W.A.: Subsonic Sting Interference on the Aerodynamic Characteristics of a Family of Slanted-Base Ogive-Cylinders. NASA CR-4299, June 1990.
10. Johnson, W.G.; Dress, D.A.: The 13-Inch Magnetic Suspension Systems Wind Tunnel. NASA TM-4090, January 1989.




Figure 1 - Beam Arrangements for 13-inch Position Sensing Systems




Figure 2 - Derivation of Extra I/O Bit from ADV-11






Figure 3 - Accommodation of Rolling Non-Axisymmetric Models (end view, LHS)
and Axially Translating Tapered Models (side view, RHS)






Figure 6 - C_p error per degree Kelvin. Standard conditions. Absolute transducer




Figure 7 - Telemetry system battery voltage sensitivity




Figure 8 - Base pressure telemetry. Typical calibration, $T = 25^{\circ}\text{C}$




Figure 9 - Base pressure telemetry. Temperature sensitivity




Figure 10 - C_p error per degree Kelvin. Standard conditions. Differential transducer

APPENDIX A
REVISED SOFTWARE LISTINGS

A.1 DIGSCN

```

;+++++
;DIGITAL INPUT PORT SCAN FOR POSITION DATA
; 4096 vertical arrays, 1024 otherwise
;
;      Input port assignments as follows:
;      Bits      Port      Channel Sensor
;      0-9       A        1      AX
;      10-15 +   A
;      0-3       B        2      FU
;      4-13      B        3      AU
;      0-11      C        4      FL
;      12-15 +   C
;      0-7       D        5      AL
;
;      Outputs to EDGE'S (CHARS 6-9) OR AX, FU, AU
;-----
;
;      .MACRO  DIGSCN
;
;      CLR      @#170466                ;MAKE SURE SIGNAL BIT CLEAR (TOP EDGE)
;
;      MOV      @#DBRA,R3                ;FETCH DIGITAL INPUT PORT DATA
;      MOV      @#DBRB,R2                ;
;      MOV      @#DBRC,R1                ;4096 ARRAYS
;      MOV      @#DBRD,R0                ;SHOULD BE TOP EDGES
;
;      MOV      #1,@#170466              ;SET SIGNAL BIT FOR LATER READ
;
;DISASSEMBLE FOR INDIVIDUAL CHANNELS
;
;      MOV      R3,R4                    ;COPY PORT A
;      BIC      #176000,R4                ;MASK HIGH BITS
;      MOV      R4,AX                    ;PUT IN AX
;
;      MOV      R3,R5                    ;COPY PORT A
;      MOV      R2,R4                    ;AND PORT B
;      ASHC     #-10.,R4                 ;JUSTIFY DATA
;      BIC      #176000,R5                ;MASK HIGH BITS
;      MOV      R5,FU                    ;PUT IN FU
;
;      MOV      R2,R4                    ;COPY PORT B
;      ASH      #-4.,R4                  ;JUSTIFY DATA
;      BIC      #176000,R4                ;MASK HIGH BITS
;      MOV      R4,AU                    ;PUT IN AU
;
;      MOV      R1,R5                    ;COPY PORT C
;      BIC      #170000,R5                ;MASK HIGH BITS
;      MOV      R5,EDGEFT                 ;PUT IN CHARS "6"

```

```

;
MOV      R0,R4                ;COPY PORT D
MOV      R1,R5                ;AND PORT C
ASHC     #-12.,R4             ;JUSTIFY DATA
BIC      #170000,R5           ;MASK HIGH BITS
MOV      R5,EDGEAT            ;PUT IN CHARS "7"
;
MOV      @#DBRC,R1            ;4096 ARRAYS
MOV      @#DBRD,R0            ;SHOULD BE BOTTOM EDGES
;
MOV      R1,R5                ;COPY PORT C
BIC      #170000,R5           ;MASK HIGH BITS
MOV      R5,EDGEFB            ;PUT IN CHARS "8"
;
MOV      R0,R4                ;COPY PORT D
MOV      R1,R5                ;AND PORT C
ASHC     #-12.,R4             ;JUSTIFY DATA
BIC      #170000,R5           ;MASK HIGH BITS
MOV      R5,EDGEAB            ;PUT IN CHARS "9"
;
CLR      @#170466              ;CLEAR SIGNAL BIT
;
.ENDM      DIGSCN

```

A.2 CNTROL - Version 1 (Modified Portion)

```
;*****
; COMPUTE MODEL POSITION AND ATTITUDE
;
;       DIGSCN                      ;SCAN DIGITAL INPUT PORT
;
; LOOKING AT THE TOP OF THE MODEL
;
;       MOV      EDGEAT,R1          ;FETCH POSITION SENSOR DATA
;       MOV      EDGEFT,R5          ;TOP EDGES
;
;       MOV      R1,AL              ;LOAD USUAL POSITION DATA LOCATIONS
;       MOV      R5,FL              ;FOR READOUTS
;
;       MOV      AU,R3              ;FETCH REST OF SENSOR DATA
;       MOV      FU,R4              ;AX=AXIAL
;       MOV      AX,R2              ;
;
;VERTICAL
;
;       MOV      R5,R0              ;FETCH FL
;       ADD      R1,R0              ;ADD AL
;       SUB      #7777,R0           ;SUBTRACT VERTICAL OFFSET (4095. PIXELS)
;       MOV      R0,VERT            ;STORE VERTICAL POSITION
;
;PITCH
;
;       MOV      R1,R0              ;FETCH AL
;       SUB      R5,R0              ;SUBTRACT FL
;       MOV      R0,PITCH           ;STORE PITCH ATTITUDE
;
;LATERAL
;
;       MOV      #1777,R0           ;FETCH LATERAL OFFSET (1023. PIXELS)
;       SUB      R3,R0              ;SUBTRACT AU
;       SUB      R4,R0              ;AND FU
;       MOV      R0,LAT             ;STORE LATERAL POSITION
;
;YAW
;
;       MOV      R3,R0              ;FETCH AU
;       SUB      R4,R0              ;SUBTRACT FU
;       MOV      R0,YAW             ;STORE YAW ATTITUDE
;
;AXIAL
;
;       MOV      R2,R0              ;FETCH AX
;       SUB      #777,R0            ;SUBTRACT AXIAL OFFSET
;       ASL      R0                 ;TIMES 2 FOR SIMILAR SENSITIVITY
;       MOV      R0,AXIAL           ;STORE AXIAL POSITION
```


A.3 CNTROL - Version 2 (Modified Portion)

```
*****
; COMPUTE MODEL POSITION AND ATTITUDE
;
;       DIGSCN                      ;SCAN DIGITAL INPUT PORT
;
; LOOKING AT THE BOTTOM OF THE MODEL
;
;       MOV      EDGEAB,R1          ;FETCH POSITION SENSOR DATA
;       MOV      EDGEFB,R5          ;BOTTOM EDGES
;
;       MOV      R1,AL              ;LOAD USUAL POSITION DATA LOCATIONS
;       MOV      R5,FL              ;FOR READOUTS
;
;       MOV      AU,R3              ;FETCH REST OF SENSOR DATA
;       MOV      FU,R4              ;AX=AXIAL
;       MOV      AX,R2              ;
;
;VERTICAL
;
;       MOV      R5,R0              ;FETCH FL
;       ADD      R1,R0              ;ADD AL
;       SUB      #7777,R0           ;SUBTRACT VERTICAL OFFSET (4095. PIXELS)
;       MOV      R0,VERT            ;STORE VERTICAL POSITION
;
;PITCH
;
;       MOV      R1,R0              ;FETCH AL
;       SUB      R5,R0              ;SUBTRACT FL
;       MOV      R0,PITCH           ;STORE PITCH ATTITUDE
;
;LATERAL
;
;       MOV      #1777,R0           ;FETCH LATERAL OFFSET (1023. PIXELS)
;       SUB      R3,R0              ;SUBTRACT AU
;       SUB      R4,R0              ;AND FU
;       MOV      R0,LAT             ;STORE LATERAL POSITION
;
;YAW
;
;       MOV      R3,R0              ;FETCH AU
;       SUB      R4,R0              ;SUBTRACT FU
;       MOV      R0,YAW             ;STORE YAW ATTITUDE
;
;AXIAL
;
;       MOV      R2,R0              ;FETCH AX
;       SUB      #777,R0            ;SUBTRACT AXIAL OFFSET
;       ASL      R0                 ;TIMES 2 FOR SIMILAR SENSITIVITY
;       MOV      R0,AXIAL           ;STORE AXIAL POSITION
```

A.4 CNTROL - Version 3 (Modified Portion)

```

;*****
; COMPUTE MODEL POSITION AND ATTITUDE
;
;         DIGSCN                                ;SCAN DIGITAL INPUT PORT
;
; LOOKING FOR BOTH EDGES ALL THE TIME
;
; START WITH BACK
;
AFTSNS:  MOV    EDGEAT,R0                        ;FETCH AT
         MOV    EDGEAB,R2                        ;FETCH AB
         CMP    R0,#16.                          ;IS MODEL HIGH?
         BMI    AHIGH                             ;JUMP IF YES
         CMP    R2,#4079.                        ;IS MODEL LOW?
         BPL    ALOW                             ;JUMP IF YES
;
; MODEL IS FULLY IN VIEW
;
ANORM:   MOV    R2,R3                            ;COPY AB
         SUB    R0,R3                            ;SUBTRACT TOP FROM BOTTOM
         ASR    R3                                ;DIVIDE WIDTH BY 2
         MOV    R3,AWIDTH                        ;STORE MODEL HALF-WIDTH
         ADD    R3,R0                            ;MODEL CENTROID LOCATION
         MOV    R0,AL                            ;STORE MODEL LOCATION
         JMP    FWDSNS                           ;DONE HERE
;
; MODEL IS HIGH
;
AHIGH:   TST    R0                               ;IS TOP EDGE IN VIEW?
         BEQ    AVHIGH                           ;JUMP IF NO
         MOV    AWIDTH,R4                        ;FETCH MODEL HALF-WIDTH
         SUB    R4,R2                            ;ADD HALF-WIDTH TO AB
         MOV    R0,R5                            ;STORE AT
         ADD    R4,R0                            ;ADD HALF-WIDTH TO AT
         SUB    R2,R0                            ;GENERATE DELTA CENTROID
         MUL    R5,R0                            ;WEIGHT DELTA
         ASR    #5,R0                            ;
         ADD    R0,R2                            ;CORRECT CENTROID
         MOV    R2,AL                            ;STORE CENTROID LOCATION
         JMP    FWDSNS                           ;DONE HERE
;
; MODEL IS VERY HIGH
;
AVHIGH:  MOV    AWIDTH,R4                        ;FETCH HALF-WIDTH
         ADD    R4,R2                            ;ADD TO AB FOR CENTROID
         MOV    R2,AL                            ;STORE CENTROID LOCATION
         JMP    FWDSNS                           ;DONE HERE
;
; MODEL IS LOW
;

```

ALOW:	TST	R2	;IS BOTTOM EDGE IN VIEW?
	BEQ	AVLOW	;JUMP IF NO
	MOV	AWIDTH,R4	;FETCH MODEL HALF-WIDTH
	NEG	R0	;SWITCH TO MEASURE FROM BOTTOM
	NEG	R2	;
	ADD	#4096.,R0	;
	ADD	#4096.,R2	;
	SUB	R4,R0	;ADD HALF-WIDTH TO AT'
	MOV	R2,R5	;STORE AB'
	ADD	R4,R2	;SUBTRACT HALF-WIDTH FROM AB'
	SUB	R0,R2	;GENERATE DELTA CENTROID
	MUL	R5,R2	;WEIGHT DELTA
	ASR	#5,R2	;
	ADD	R2,R0	;CORRECT CENTROID
	NEG	R0	;RETURN TO MEASURE FROM TOP
	ADD	#4096.,R0	;
	MOV	R0,AL	;STORE
	JMP	FWDSNS	;DONE HERE
;			
; MODEL IS VERY LOW			
;			
AVLOW:	MOV	AWIDTH,R4	;FETCH MODEL HALF-WIDTH
	SUB	R4,R0	;SUBTRACT FROM AT
	MOV	R0,AL	;STORE
	JMP	FWDSNS	;DONE HERE
;			
; NOW DO FRONT			
;			
FWDSNS:	MOV	EDGEFT,R0	;FETCH FT
	MOV	EDGEFB,R2	;FETCH FB
	CMP	R0,#16.	;IS MODEL HIGH?
	BMI	FHIGH	;JUMP IF YES
	CMP	R2,#4079.	;IS MODEL LOW?
	BPL	FLOW	;JUMP IF YES
;			
; MODEL IS FULLY IN VIEW			
;			
FNORM:	MOV	R2,R3	;COPY FB
	SUB	R0,R3	;SUBTRACT TOP FROM BOTTOM
	ASR	R3	;DIVIDE WIDTH BY 2
	MOV	R3,FWIDTH	;STORE MODEL HALF-WIDTH
	ADD	R3,R0	;MODEL CENTROID LOCATION
	MOV	R0,FL	;STORE MODEL LOCATION
	JMP	SNSDN	;DONE HERE
;			
; MODEL IS HIGH			
;			
FHIGH:	TST	R0	;IS TOP EDGE IN VIEW?
	BEQ	FVHIGH	;JUMP IF NO
	MOV	FWIDTH,R4	;FETCH MODEL HALF-WIDTH
	SUB	R4,R2	;ADD HALF-WIDTH TO FB
	MOV	R0,R5	;STORE FT
	ADD	R4,R0	;ADD HALF-WIDTH TO FT
	SUB	R2,R0	;GENERATE DELTA CENROID

MUL	R5,R0	;WEIGHT DELTA
ASR	#5,R0	;
ADD	R0,R2	;CORRECT CENTROID
MOV	R2,FL	;STORE CENTROID LOCATION
JMP	SNSDN	;DONE HERE
;		
; MODEL IS VERY HIGH		
;		
FVHIGH:	MOV FWIDTH,R4	;FETCH HALF-WIDTH
	ADD R4,R2	;ADD TO FB FOR CENTROID
	MOV R2,FL	;STORE CENTROID LOCATION
	JMP SNSDN	;DONE HERE
;		
; MODEL IS LOW		
;		
FLOW:	TST R2	;IS BOTTOM EDGE IN VIEW?
	BEQ FVLOW	;JUMP IF NO
	MOV FWIDTH,R4	;FETCH MODEL HALF-WIDTH
	NEG R0	;SWITCH TO MEASURE FROM BOTTOM
	NEG R2	;
	ADD #4096.,R0	;
	ADD #4096.,R2	;
	SUB R4,R0	;ADD HALF-WIDTH TO FT'
	MOV R2,R5	;STORE FB'
	ADD R4,R2	;SUBTRACT HALF-WIDTH FROM FB'
	SUB R0,R2	;GENERATE DELTA CENTROID
	MUL R5,R2	;WEIGHT DELTA
	ASR #5,R2	;
	ADD R2,R0	;CORRECT CENTROID
	NEG R0	;RETURN TO MEASURE FROM TOP
	ADD #4096.,R0	;
	MOV R0,FL	;STORE
	JMP SNSDN	;DONE HERE
;		
; MODEL IS VERY LOW		
;		
FVLOW:	MOV FWIDTH,R4	;FETCH MODEL HALF-WIDTH
	SUB R4,R0	;SUBTRACT FROM FT
	MOV R0,FL	;STORE
SNSDN:		